
NAG Library Routine Document

E02BAF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

E02BAF computes a weighted least squares approximation to an arbitrary set of data points by a cubic
spline with knots prescribed by you. Cubic spline interpolation can also be carried out.

2 Specification

SUBROUTINE E02BAF (M, NCAP7, X, Y, W, LAMDA, WORK1, WORK2, C, SS, IFAIL)

INTEGER M, NCAP7, IFAIL

REAL (KIND=nag_wp) X(M), Y(M), W(M), LAMDA(NCAP7), WORK1(M),
WORK2(4*NCAP7), C(NCAP7), SS

&

3 Description

E02BAF determines a least squares cubic spline approximation s xð Þ to the set of data points xr; yrð Þ with
weights wr, for r ¼ 1; 2; . . . ;m. The value of NCAP7 ¼ �nþ 7, where �n is the number of intervals of the
spline (one greater than the number of interior knots), and the values of the knots �5; �6; . . . ; ��nþ3, interior
to the data interval, are prescribed by you.

s xð Þ has the property that it minimizes �, the sum of squares of the weighted residuals �r, for
r ¼ 1; 2; . . . ;m, where

�r ¼ wr yr � s xrð Þð Þ.
The routine produces this minimizing value of � and the coefficients c1; c2; . . . ; cq, where q ¼ �nþ 3, in the
B-spline representation

s xð Þ ¼
Xq

i¼1

ciNi xð Þ.

Here Ni xð Þ denotes the normalized B-spline of degree 3 defined upon the knots �i; �iþ1; . . . ; �iþ4.

In order to define the full set of B-splines required, eight additional knots �1; �2; �3; �4 and
��nþ4; ��nþ5; ��nþ6; ��nþ7 are inserted automatically by the routine. The first four of these are set equal to
the smallest xr and the last four to the largest xr.

The representation of s xð Þ in terms of B-splines is the most compact form possible in that only �nþ 3
coefficients, in addition to the �nþ 7 knots, fully define s xð Þ.
The method employed involves forming and then computing the least squares solution of a set of m linear
equations in the coefficients ci, for i ¼ 1; 2; . . . ; �nþ 3. The equations are formed using a recurrence
relation for B-splines that is unconditionally stable (see Cox (1972) and de Boor (1972)), even for multiple
(coincident) knots. The least squares solution is also obtained in a stable manner by using orthogonal
transformations, viz. a variant of Givens rotations (see Gentleman (1974) and Gentleman (1973)). This
requires only one equation to be stored at a time. Full advantage is taken of the structure of the equations,
there being at most four nonzero values of Ni xð Þ for any value of x and hence at most four coefficients in
each equation.

For further details of the algorithm and its use see Cox (1974), Cox (1975) and Cox and Hayes (1973).

Subsequent evaluation of s xð Þ from its B-spline representation may be carried out using E02BBF. If
derivatives of s xð Þ are also required, E02BCF may be used. E02BDF can be used to compute the definite
integral of s xð Þ.

5 Parameters

1: M – INTEGER Input

On entry: the number m of data points.

Constraint: M � mdist � 4, where mdist is the number of distinct x values in the data.

2: NCAP7 – INTEGER Input

On entry: �nþ 7, where �n is the number of intervals of the spline (which is one greater than the
number of interior knots, i.e., the knots strictly within the range x1 to xm) over which the spline is
defined.

Constraint: 8 � NCAP7 � mdistþ 4, where mdist is the number of distinct x values in the data.

3: XðMÞ – REAL (KIND=nag_wp) array Input

On entry: the values xr of the independent variable (abscissa), for r ¼ 1; 2; . . . ;m.

Constraint: x1 � x2 � � � � � xm.

4: YðMÞ – REAL (KIND=nag_wp) array Input

On entry: the values yr of the dependent variable (ordinate), for r ¼ 1; 2; . . . ;m.

5: WðMÞ – REAL (KIND=nag_wp) array Input

On entry: the values wr of the weights, for r ¼ 1; 2; . . . ;m. For advice on the choice of weights,
see the E02 Chapter Introduction.

Constraint: WðrÞ > 0:0, for r ¼ 1; 2; . . . ;m.

6: LAMDAðNCAP7Þ – REAL (KIND=nag_wp) array Input/Output

On entry: LAMDAðiÞ must be set to the i� 4ð Þth (interior) knot, �i, for i ¼ 5; 6; . . . ; �nþ 3.

Constraint: Xð1Þ < LAMDAð5Þ � LAMDAð6Þ � � � � � LAMDAðNCAP7� 4Þ < XðMÞ.
On exit: the input values are unchanged, and LAMDAðiÞ, for i ¼ 1; 2; 3; 4, NCAP7� 3,
NCAP7� 2, NCAP7� 1, NCAP7 contains the additional (exterior) knots introduced by the
routine. For advice on the choice of knots, see Section 3.3 in the E02 Chapter Introduction.

7: WORK1ðMÞ – REAL (KIND=nag_wp) array Workspace
8: WORK2ð4� NCAP7Þ – REAL (KIND=nag_wp) array Workspace

9: CðNCAP7Þ – REAL (KIND=nag_wp) array Output

On exit: the coefficient ci of the B-spline Ni xð Þ, for i ¼ 1; 2; . . . ; �nþ 3. The remaining elements of
the array are not used.

10: SS – REAL (KIND=nag_wp) Output

On exit: the residual sum of squares, �.

11: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

The knots fail to satisfy the condition

Xð1Þ < LAMDAð5Þ � LAMDAð6Þ � � � � � LAMDAðNCAP7� 4Þ < XðMÞ.
Thus the knots are not in correct order or are not interior to the data interval.

IFAIL ¼ 2

The weights are not all strictly positive.

IFAIL ¼ 3

The values of XðrÞ, for r ¼ 1; 2; . . . ;M, are not in nondecreasing order.

IFAIL ¼ 4

NCAP7 < 8 (so the number of interior knots is negative) or NCAP7 > mdistþ 4, where mdist is
the number of distinct x values in the data (so there cannot be a unique solution).

IFAIL ¼ 5

The conditions specified by Schoenberg and Whitney (1953) fail to hold for at least one subset of
the distinct data abscissae. That is, there is no subset of NCAP7� 4 strictly increasing values,
X R 1ð Þð Þ;X R 2ð Þð Þ; . . . ;X R NCAP7� 4ð Þð Þ, among the abscissae such that

X R 1ð Þð Þ < LAMDAð1Þ < X R 5ð Þð Þ,
X R 2ð Þð Þ < LAMDAð2Þ < X R 6ð Þð Þ,

X R NCAP7� 8ð Þð Þ < LAMDAðNCAP7� 8Þ < X R NCAP7� 4ð Þð Þ.
This means that there is no unique solution: there are regions containing too many knots compared
with the number of data points.

7 Accuracy

The rounding errors committed are such that the computed coefficients are exact for a slightly perturbed set
of ordinates yr þ �yr. The ratio of the root-mean-square value for the �yr to the root-mean-square value of
the yr can be expected to be less than a small multiple of ��m�machine precision, where � is a
condition number for the problem. Values of � for 20–30 practical datasets all proved to lie between 4:5
and 7:8 (see Cox (1975)). (Note that for these datasets, replacing the coincident end knots at the end
points x1 and xm used in the routine by various choices of non-coincident exterior knots gave values of �
between 16 and 180. Again see Cox (1975) for further details.) In general we would not expect � to be
large unless the choice of knots results in near-violation of the Schoenberg–Whitney conditions.

A cubic spline which adequately fits the data and is free from spurious oscillations is more likely to be
obtained if the knots are chosen to be grouped more closely in regions where the function (underlying the
data) or its derivatives change more rapidly than elsewhere.

8 Further Comments

The time taken is approximately C � 2mþ �nþ 7ð Þ seconds, where C is a machine-dependent constant.

Multiple knots are permitted as long as their multiplicity does not exceed 4, i.e., the complete set of knots
must satisfy �i < �iþ4, for i ¼ 1; 2; . . . ; �nþ 3, (see Section 6). At a knot of multiplicity one (the usual
case), s xð Þ and its first two derivatives are continuous. At a knot of multiplicity two, s xð Þ and its first
derivative are continuous. At a knot of multiplicity three, s xð Þ is continuous, and at a knot of multiplicity
four, s xð Þ is generally discontinuous.

The routine can be used efficiently for cubic spline interpolation, i.e., if m ¼ �nþ 3. The abscissae must
then of course satisfy x1 < x2 < � � � < xm. Recommended values for the knots in this case are �i ¼ xi�2,
for i ¼ 5; 6; . . . ; �nþ 3.

9 Example

Determine a weighted least squares cubic spline approximation with five intervals (four interior knots) to a
set of 14 given data points. Tabulate the data and the corresponding values of the approximating spline,
together with the residual errors, and also the values of the approximating spline at points half-way
between each pair of adjacent data points.

The example program is written in a general form that will enable a cubic spline approximation with �n
intervals (�n� 1 interior knots) to be obtained to m data points, with arbitrary positive weights, and the
approximation to be tabulated. Note that E02BBF is used to evaluate the approximating spline. The
program is self-starting in that any number of datasets can be supplied.

9.1 Program Text

Program e02bafe

! E02BAF Example Program Text

! Mark 24 Release. NAG Copyright 2012.

! .. Use Statements ..
Use nag_library, Only: e02baf, e02bbf, nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..

Real (Kind=nag_wp) :: fit, ss, xarg
Integer :: ifail, iwght, j, m, ncap, ncap7, r

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: c(:), lamda(:), w(:), work1(:), &

work2(:), x(:), y(:)
! .. Executable Statements ..

Write (nout,*) ’E02BAF Example Program Results’

! Skip heading in data file
Read (nin,*)

Read (nin,*) m
Read (nin,*) ncap, iwght
ncap7 = ncap + 7
Allocate (x(m),y(m),w(m),lamda(ncap7),c(ncap7),work1(m),work2(4*ncap7))

Read (nin,*) lamda(5:(ncap+3))

Do r = 1, m

If (iwght==1) Then
Read (nin,*) x(r), y(r)
w(r) = 1.0E0_nag_wp

Else
Read (nin,*) x(r), y(r), w(r)

End If

End Do

ifail = 0
Call e02baf(m,ncap7,x,y,w,lamda,work1,work2,c,ss,ifail)

Write (nout,*)
Write (nout,*) ’ J LAMDA(J+2) B-spline coeff C(J)’
Write (nout,*)

j = 1
Write (nout,99998) j, c(1)

Do j = 2, ncap + 2
Write (nout,99999) j, lamda(j+2), c(j)

End Do

Write (nout,99998) ncap + 3, c(ncap+3)
Write (nout,*)
Write (nout,99997) ’Residual sum of squares = ’, ss
Write (nout,*)
Write (nout,*) ’Cubic spline approximation and residuals’
Write (nout,*)
Write (nout,*) &

’ R Abscissa Weight Ordinate Spline Residual’
Write (nout,*)

Do r = 1, m

ifail = 0
Call e02bbf(ncap7,lamda,c,x(r),fit,ifail)

Write (nout,99995) r, x(r), w(r), y(r), fit, fit - y(r)

If (r<m) Then
xarg = 0.5E0_nag_wp*(x(r)+x(r+1))

ifail = 0
Call e02bbf(ncap7,lamda,c,xarg,fit,ifail)

Write (nout,99996) xarg, fit
End If

End Do

99999 Format (1X,I3,F15.4,F20.4)
99998 Format (1X,I3,F35.4)
99997 Format (1X,A,E12.2)
99996 Format (1X,F14.4,F33.4)
99995 Format (1X,I3,4F11.4,E10.2)

End Program e02bafe

9.2 Program Data

E02BAF Example Program Data
14
5 2

1.50
2.60
4.00
8.00
0.20 0.00 0.20
0.47 2.00 0.20
0.74 4.00 0.30
1.09 6.00 0.70
1.60 8.00 0.90
1.90 8.62 1.00
2.60 9.10 1.00
3.10 8.90 1.00
4.00 8.15 0.80
5.15 7.00 0.50
6.17 6.00 0.70
8.00 4.54 1.00

10.00 3.39 1.00
12.00 2.56 1.00

9.3 Program Results

E02BAF Example Program Results

J LAMDA(J+2) B-spline coeff C(J)

1 -0.0465
2 0.2000 3.6150
3 1.5000 8.5724
4 2.6000 9.4261
5 4.0000 7.2716
6 8.0000 4.1207
7 12.0000 3.0822
8 2.5597

Residual sum of squares = 0.18E-02

Cubic spline approximation and residuals

R Abscissa Weight Ordinate Spline Residual

1 0.2000 0.2000 0.0000 -0.0465 -0.47E-01
0.3350 1.0622

2 0.4700 0.2000 2.0000 2.1057 0.11E+00
0.6050 3.0817

3 0.7400 0.3000 4.0000 3.9880 -0.12E-01
0.9150 5.0558

4 1.0900 0.7000 6.0000 5.9983 -0.17E-02
1.3450 7.1376

5 1.6000 0.9000 8.0000 7.9872 -0.13E-01
1.7500 8.3544

6 1.9000 1.0000 8.6200 8.6348 0.15E-01
2.2500 9.0076

7 2.6000 1.0000 9.1000 9.0896 -0.10E-01
2.8500 9.0353

8 3.1000 1.0000 8.9000 8.9125 0.12E-01
3.5500 8.5660

9 4.0000 0.8000 8.1500 8.1321 -0.18E-01
4.5750 7.5592

10 5.1500 0.5000 7.0000 6.9925 -0.75E-02
5.6600 6.5010

11 6.1700 0.7000 6.0000 6.0255 0.26E-01
7.0850 5.2292

12 8.0000 1.0000 4.5400 4.5315 -0.85E-02
9.0000 3.9045

13 10.0000 1.0000 3.3900 3.3928 0.28E-02
11.0000 2.9574

14 12.0000 1.0000 2.5600 2.5597 -0.35E-03

Weighted Least-squares Cubic Spline Approximation to a Set of 14 Data Points

